Homogeneous [**collisionless**] sheath model

```
(*Integrage the electric field to get the space potential*)
Clear[e, n, ϵ0, sa]
```

$$\texttt{Integrate}\left[\frac{e\,n}{\epsilon 0}\,(xp - sa[t]),\ \{xp,\ x,\ sa[t]\}\right]$$

$$-\frac{e\,n\,x^2}{2\,\epsilon 0} + \frac{e\,n\,x\,sa[t]}{\epsilon 0} - \frac{e\,n\,sa[t]^2}{2\,\epsilon 0}$$

```
(*solve the equation to find out the system current
 required to get a given max voltage across the sheath*)
Clear[e, n, ω, ϵ0, J, Vmax]
```

$$\texttt{Solve}\left[\frac{2}{\epsilon 0}\,\frac{J^2}{e\,n\,\omega^2} == Vmax,\ J\right]$$

$$\left\{\left\{J \to -\frac{\sqrt{e}\,\sqrt{n}\,\sqrt{Vmax}\,\sqrt{\epsilon 0}\,\omega}{\sqrt{2}}\right\},\ \left\{J \to \frac{\sqrt{e}\,\sqrt{n}\,\sqrt{Vmax}\,\sqrt{\epsilon 0}\,\omega}{\sqrt{2}}\right\}\right\}$$

```
Clear[s0, ω, ϵ0, n, e]
sa[t_] := s0 - s0 Sin[ω t]
```

$$Vpa[t\_] := -\frac{e\,n}{2\,\epsilon 0}\,s0^2\,(1 - Sin[\omega\,t])^2$$

$$-\frac{e\,n\,x^2}{2\,\epsilon 0} + \frac{e\,n\,x\,sa[t]}{\epsilon 0} - \frac{e\,n\,sa[t]^2}{2\,\epsilon 0}\ \ //\ \texttt{FullSimplify}$$

```
-Vpa[t]
```

$$-\frac{e\,n\,(-s0 + x + s0\,Sin[t\,\omega])^2}{2\,\epsilon 0}$$

$$\frac{e\,n\,s0^2\,(1 - Sin[t\,\omega])^2}{2\,\epsilon 0}$$

$$f = 13.56 * 10^{6}; \; (*\text{Hz}*) \;\; (***\text{NEW PARAMETER FOR THE MODEL}*****)$$

$$T = 1 / f; \;\; (***\text{NEW PARAMETER FOR THE MODEL}*****)$$

$$\epsilon 0 = 8.8541878128 * 10^{-12};$$

$$e = 1.60217662 * 10^{-19}; (*\text{C}*)$$

$$n = 10^{16}; \; (*\text{m}-3*)$$

$$\omega = 2 \pi f;$$

```
Vmax = 1000; (**INPUT THIS AND CALCULATE J from it****)
```

$$J = \omega \sqrt{\frac{e\, n\, \epsilon 0\, \text{Vmax}}{2}} \; ;(*\text{A/m2}*) \; (***\text{NEW PARAMETER FOR THE MODEL the total current through the discharge} *****)$$

$$s0 = \frac{J}{e\, n\, \omega};$$

```
smean = s0;

(*The sheath size varies*)
sa[t_] := smean - s0 Sin[ω t]

(*The space potential across the sheath*)
(*The max voltage difference is when Sinwt is -1*)
```

$$\text{Vpa}[t\_] := -\frac{e\, n}{2\, \epsilon 0}\, s0^{2}\, (1 - \text{Sin}[\omega\, t])^{2}$$

$$\text{Vpa}\left[\frac{3}{4}\, T\right]$$

```
(*The space potential in x and time;
subtract Vpa at x = 0 to get the voltage referenced to the electrode at x=0*)
```

$$\text{Vpa}[x\_, t\_] := \text{If}\left[x \le \text{sa}[t], -\frac{e\, n\, x^{2}}{2\, \epsilon 0} + \frac{e\, n\, x\, \text{sa}[t]}{\epsilon 0} - \frac{e\, n\, \text{sa}[t]^{2}}{2\, \epsilon 0}, 0\right] - \text{Vpa}[t]$$

```
(*The electric field in x and time*)
```

$$\text{Efield}[x\_, t\_] := \text{If}\left[x \le \text{sa}[t], \frac{e\, n}{\epsilon 0}\, (x - \text{sa}[t]), 0\right]$$

```
-1000.

Plot[Vpa[T * τ], {τ, 0, 1},
  FrameLabel → {"time (x1 period)", "Vpa (V)"}, Frame → True];
```

```
Plot[10^2 * sa[T * τ], {τ, 0, 1},
 FrameLabel → {"time (x1 period)", "Sheath Size (cm)"}, Frame → True]
10^2 * sa[T * 3/4] (*cm*)
```




```
0.332456
```

```
Plot[Efield[0, T * τ], {τ, 0, 1}, FrameLabel → {"time (x1 period)", "E (V/m)"},
  Frame → True];
```

```
τs = Random[] * T
Plot3D[Vpa[x * 10^-2, τ + τs], {x, 0, 2 * s0 * 10^2}, {τ, 0, T},
 AxesLabel → {"x (cm)", "time (x 1T)", "Vpa (V)"}, PlotRange → All]
```

3.34417×10^{-9}



```
Plot3D[Efield[x * 10^-2, τ * T], {x, 0, 2 * s0 * 10^2},
 {τ, 0, 1}, AxesLabel → {"x (cm)", "time (x 1T)", "E (V/m)"}]
```




```
Plot3D[Vpa[x * 10^-2, τ * T] + Vpa[τ T], {x, 0, 2 * s0 * 10^2}, {τ, 0, 1},
 AxesLabel → {"x (cm)", "time (x 1T)", "Vpa (V)"}, PlotRange → All]
```



Inhomogeneous **collisionless** sheath model

Clear all constants, variables and functions

```
Clear[ϵ0, e, f, T, ω, Te, pPa, Vpkpk, pmTorr, λi, s0, λD, H, n,
 J, sm, x, τ, Potential, Ez, ϕx, integral, interpolationData, s]
```

Physics Constants

```
ϵ0 = 8.8541878128 * 10^-12;
e = 1.60217662 * 10^-19; (*C*)
```

Collisionless solution, Lieberman, 1988

The plasma parameters are input to the model:
* discharge frequency,
* electron temperature, and
* voltage pk-pk across the sheath

```
f = 13.56 * 10^6; (*Hz*)    T = 1 / f;    ω = 2 π f;
Te = 3; (*eV*)
Vpkpk = 2000; (*V*)
```

The effective oscillation parameter s0 and the Debye length formulas are as shown below (these expressions are the same for collisionless and collisional solutions).

$$s0 = \frac{J}{e\, n\, \omega};$$

$$\lambda D = \sqrt{\frac{\epsilon 0\, Te}{e\, n}};$$

In the collisionless solution, the H parameter is given below.

$$H = \frac{1}{\pi}\frac{s0^2}{\lambda D^2}; \ (*\ \text{collisionless}\ *)$$

From the inhomogeneous model, we want to estimate the size of the sheath which is larger than in the hom. model. Then from equation (11.2.28) Lieberman's book, we solve x for the $\phi = \pi$.

Estimating the Plasma Discharge Current Density amplitude (J) from the plasma parameters

The expression to estimate the plasma discharge current density amplitude (J), given the plasma parameters,

$$n = 10^{16}; \ (*m^{-3}*)$$

$$J = \frac{2\,\omega}{5}\sqrt{\frac{2}{5}}\sqrt{e\, n\, \epsilon 0}\sqrt{\sqrt{3\, Te}\sqrt{192\, Te + 125\, Vpkpk} - 24\, Te} \ (*A/m2*)$$

97.0701

ALTERNATIVE : Estimating the Plasma edge Density (N) from the plasma parameters

Alternatively, using the expression above we can determine the plasma density at the sheath edge (n) as a function of the plasma parameters,

$$J = 25; \ (*A/m2*)$$

$$n = \frac{125\, J^2}{8\, e\sqrt{Te}\left(-24\sqrt{Te} + \sqrt{3}\sqrt{192\, Te + 125\, Vpkpk}\right)\epsilon 0\, \omega^2} \ (*m^{-3}*)$$

6.63298×10^{14}

The sheath edge (s(t))

The sheath edge s(t) is determined by the function below. Note that the function is valid for π in the range 0 to π. The solution is mirrored for the -π to 0 range. The parameters that determine the sheath size are the current through the plasma discharge (J), the plasma density n, the driving frequency (ω), and the Debye length (electron temperature and plasma density).

$$x[\phi\_] := s0 \left((1 - Cos[\phi]) + \frac{H}{8} \left(\frac{3}{2} Sin[\phi] + \frac{11}{18} Sin[3\,\phi] - 3\,\phi\, Cos[\phi] - \frac{1}{3}\,\phi\, Cos[3\,\phi] \right) \right)$$

```
τ[t_] := Mod[t + T / 2, T] - T / 2
s[t_] := x[Abs[ω τ[t]]]
```

```
Plot[s[t T] * 100, {t, - 1 / 2, 1 / 2}, PlotRange → All, Frame → True, FrameLabel →
  {"time (*T)", "Position of sheath edge\n from the plasma, s[t] (cm)"}]
```



The sheath size when fully extended is the value of x at ϕ=π, or, the same s at time t=T/2,

```
sm = x[π]
sm = s[T / 2]
```

```
0.0104595
```

```
0.0104595
```

In the model of the present document, we change from the variable x to the variable z; where z is the distance from the electrode; i.e. z=0 is at the electrode and z = sm is the fully extended sheath edge.

The sheath positiion z as a function of time (z = sm - x). Using meter and second units.

```
z[t_] := s[T / 2] - s[t]
Plot[z[t], {t, - T, T}, PlotRange → All,
 Frame → True, FrameLabel → {"time (s)", "z(t) (m)"}]
```



Inverse of function x(ϕ): ϕ(x)

The inverse of the ϕ(x) function is required to calculate the electric field.

```
npoints = 64;
numData = Table[{x[A], A}, {A, 0, π, π / npoints}];
ϕx = Interpolation[%];

Show[
 {Plot[ϕx[xxx], {xxx, 0, sm}, Frame → True],
  ListPlot[numData]}
 , Frame → True, FrameLabel → {"x (elect 0; to plasma sm)", "ϕ(x)"}]
```



Electric Field in the sheath (z,t)

From Lieberman, 1988. Equation (28). where in the model of this document the relationship between the x variable and z variable is: x = sm - z. Therefore the electric field is not null for the condtion that s(t) < x = sm-z.

```
Ef[xx_, t_] := If[s[t] < xx, J/(ϵ0 ω) (Cos[ω t] - Cos[ϕx[xx]]), 0]
Ez[zz_, t_] := -Ef[sm - zz, t]
```

The graph below shows a plot of the electric field in the sheath from time -T/2 to time T/2,

The plot shows that the electric field is approximately zero where the position is behind the sheath edge as defined in the function.

```
Plot3D[Ez[zz / 100, tt * T], {zz, 0, 100 * sm}, {tt, -1 / 2, 1 / 2}, PlotRange → All,
 AxesLabel → {"z (m)", "t (* T)", "Ef (V/m)"}, Mesh → False, PlotPoints → {25, 25}]
```



Potential across the sheath (z,t)

The voltage or space potential across the sheath as a function of time can be obtained by integrating the electric field function as follows.

```
Potential[zz_, t_] := -NIntegrate[Ez[zzz, t],
   {zzz, 0, zz}, Method → "TrapezoidalRule", PrecisionGoal → 2]
```

The voltage integration is computationally expensive. Therefore, to illustrate the potential, we first generate a data set.

```
potential3D =
  Table[Potential[zz, tt], {zz, 0, sm, sm / 10}, {tt, -T / 2, T / 2, T / 25}];
```

And then plot the data using ListPlot3D.

The plot shows that the voltage is zero at the electrode position (z=0) and that is modulates in the sheath as a function of time, where the maximum voltage difference across the sheath occurs at t=0 (remember that in the model of this document we have changed from x variable to z and that the sheath is fully extended at time t = 0).

```
ListPlot3D[potential3D, AxesLabel → {"t (au)", "z (au)", "P (V)"}, Mesh → False]
```



The space potential can be also plotted as single curves for given time values. The graph below shows said curves for t from 0 to T in increments of T/10.

```
ListPlot[
 Table[
  Table[{zz * 100, Potential[zz, tt]}, {zz, 0, sm, sm / 10}],
  {tt, 0, T / 2, T / 10}], Joined → True,
 Frame → True, FrameLabel → {"z (cm)", "Potential (V)"}]
```



Note that in this model, the plasma parameters including the maximum voltage across the sheath Vpkpk were used as input to define the last parameter, the plasma discharge current density J. Therefore, when the sheath is fully extended, at z=sm and t=0 the voltage difference should be equal to Vpkpk.

This is calculated below; however, due to simplified numerical calculations made to reduce computational time, the expected potential differs from the input potential by a couple of percent.

```
Vpkpk
Potential[sm, 0]
Abs[(Vpkpk - Potential[sm, 0])/Vpkpk] * 100 "%"
```

```
2000
2000.03
0.00130912 %
```

Inhomogeneous **collisional** sheath model

Clear all constants, variables and functions

```
Clear[ϵ0, e, f, T, ω, Te, pPa, Vpkpk, pmTorr, λi, s0, λD, H, n,
 J, sm, x, τ, Potential, Ez, ϕx, integral, interpolationData, s]
```

Physics Constants

```
ϵ0 = 8.8541878128 * 10^-12;
e = 1.60217662 * 10^-19; (*C*)
```

Collisional solution

The plasma parameters are input to the model:
* discharge frequency,
* electron temperature, and
* neutral gas pressure.
* voltage pk-pk across the sheath

```
f = 13.56 * 10^6; (*Hz*)    T = 1 / f; (*s*)    ω = 2 π f;
Te = 3; (*eV*)
pPa = 0.5; (*Pascal*)
Vpkpk = 2000; (*V*)
```

The ion mean free path in this model is estimated by the formula below. Note that the pressure in milliTorr in the formula below to estimate the ion mean free path. This formula is valid for argon, see Lieberman 1989 publication.

```
pmTorr = pPa / 0.13332237;
λi = (30 * pmTorr)^-1 (*m*) // N;
```

The effective oscillation parameter s0 and the Debye length formulas are still the same as in the collisionless solution.

```
s0 = J/(e n ω);
λD = Sqrt[(ϵ0 Te)/(e n)];
```

In the collisional solution, the H parameter is given below. Where the ion mean free path is assumed as a constant..

$$H = \left(\frac{2\,\lambda i}{\pi^2}\ \frac{s0}{\lambda D^2}\right)^{1/2}; \text{ (* collisional *)}$$

Estimating the Plasma Discharge Current Density amplitude (J) from the plasma parameters

The Voltage across the sheath, when the sheath is fully expanded is V(t=0). Using equation (25) in Lieberman, 1989, the following integral between 0 and π should give approximately 2.5; which is the result of the integral below when solving equation (25). The result of the integral is not exactly 2.5 but it is a good approximation.

- $\text{NIntegrate}\left[(1 - \text{Cos}[\zeta])\ (\text{Sin}[\zeta] - \zeta\,\text{Cos}[\zeta])^{1/2}\ \text{Sin}[\zeta],\ \{\zeta, 0, \pi\}\right]$
- 2.499860549870389`

Following from this, we use equation (26) in Lieberman, 1989, to get an analytical expression for J given the other plasma parameters.

- Clear[H, s0, J, ∈0, ω, Vpkpk, e, n, Te, λD, λi]

$$s0 = \frac{J}{e\,n\,\omega};$$

$$\lambda D = \sqrt{\frac{\epsilon 0\ Te}{e\,n}};$$

$$H = \left(\frac{2\,\lambda i}{\pi^2}\ \frac{s0}{\lambda D^2}\right)^{1/2}; \text{ (* collisional *)}$$

$$sol = \text{Solve}\left[Vpkpk == \frac{2.5\,H\,e\,n\,s0^2}{\epsilon 0},\ J\right];$$

sol[[5]]

- $$\left\{J \to \frac{0.953845\ e^{2/5}\ n^{2/5}\ Te^{1/5}\ Vpkpk^{2/5}\ \epsilon 0^{3/5}\ \omega}{\lambda i^{1/5}}\right\}$$

Therefore, using the expression above, the Plasma Discharge Current Density Amplitude is estimated from the other parameters input to the model as per the following formula. The missing parameter to specify was the plasma density at the edge of the sheath (n).

$$\mathbf{n = 10^{16};\ (*m^{-3}*)}$$

$$\mathbf{J = 0.9538453052369402\ \omega\left(\frac{e^2\ n^2\ Te\ Vpkpk^2\ \epsilon 0^3}{\lambda i}\right)^{1/5};\ (*A/m^2*)}$$

J

96.8645

ALTERNATIVE : Estimating the Plasma edge Density (N) from the plasma parameters

Similarly to the estimate of the current density, here we give the current density as an input and estimate the plasma density at the edge of the sheath. Following from this, we use equation (26) in

Lieberman, 1989, to get an analytical expression for n given the other plasma parameters.

```
Clear[H, s0, J, ∈0, ω, Vpkpk, e, n, Te, λD, λi]
```

$$s0 = \frac{J}{e\, n\, \omega};$$

$$\lambda D = \sqrt{\frac{\epsilon 0\, Te}{e\, n}};$$

$$H = \left(\frac{2\, \lambda i}{\pi^2}\, \frac{s0}{\lambda D^2}\right)^{1/2}; \text{ (* collisional *)}$$

$$sol = Solve\left[Vpkpk == \frac{2.5\, H\, e\, n\, s0^2}{\epsilon 0}, n\right];$$

```
sol[[1]]
```

$$\left\{n \to \frac{1.1254\, J^2 \sqrt{\frac{J\, \lambda i}{Te\, \epsilon 0\, \omega}}}{e\, Vpkpk\, \epsilon 0\, \omega^2}\right\}$$

Therefore, using the expression above, the Plasma Density is estimated from the other parameters input to the model as per the following formula.

J = 10; (*A/m2*)

$$n = \frac{1.1253953951963827\text{`}\, J^2 \sqrt{\frac{J\, \lambda i}{Te\, \epsilon 0\, \omega}}}{e\, Vpkpk\, \epsilon 0\, \omega^2}; \text{ (*}m^{-3}\text{*)}$$

n

1.71221×10^{13}

The sheath edge (s(t))

The sheath edge, i.e., the x coordinate at which the electron density drops in a step-like fashion, is estimated numerically, from Lieberman 1989, equation (14),

```
x[ϕ_] := s0 * H * NIntegrate[(Sin[ξ] - ξ Cos[ξ])^(1/2) Sin[ξ] , {ξ, 0, ϕ}]
τ[t_] := Mod[t + T / 2, T] - T / 2
s[t_] := x[Abs[ω τ[t]]]

Plot[s[t T] * 100, {t, - 1 / 2, 1 / 2}, PlotRange → All, Frame → True, FrameLabel →
  {"time (*T)", "Position of sheath edge\n from the plasma, s[t] (cm)"}]
```



The numerical integration in the expression above is computationally expensive. In addition, it is noted that the integration is independent of the plasma parameters. For instance, we can compute the integral in the range 0 to π over a uniform grid and use a linear interpolation function to enable faster computing.

A better numerical solution is to do a proper numerical integration of the integral function. Below we do that calculation for the selected ϕ values and the numerical approximation to the integral (addition of terms multiplied by the step size in ϕ). In the plot below it is hard to see any substantial difference, but the small differences may add up and when the voltage across the sheath is calculated there is a more than 1% error. Hence, recommend to integrate func of ϕ using a proper algorithm for integration; for example a simple trapezoidal rule is sufficient.

A straight numerical integration using $2^N + 1$ values

```
integralfunc[ϕ_] :=
 NIntegrate[(Sin[ξ] - ξ Cos[ξ])^(1/2) Sin[ξ] , {ξ, 0, ϕ}, Method → "TrapezoidalRule"]
```

The full list of ζ and integral values is in the table below which is used as input to a linear interpolation function. In this case we are using data from the proper integration of the function of ϕ.

```
NPoints = 2^6 + 1;   Δϕ = π / (NPoints - 1);
zetas = Table[ξ, {ξ, 0, π, Δϕ}];
integralfuncdata = Table[integralfunc[zetas[[i]]], {i, 1, Length[zetas]}]

interpolationData = Transpose[{zetas, integralfuncdata}];
```

```
{0., 4.32137×10^-6, 0.000048842, 0.000201554, 0.000550385, 0.00119826,
 0.00225994, 0.00385949, 0.00612818, 0.00920244, 0.0132221, 0.0183285,
 0.024663, 0.032365, 0.0415709, 0.052412, 0.0650132, 0.079492, 0.0959565,
 0.114505, 0.135223, 0.158185, 0.18345, 0.211066, 0.241061, 0.27345,
 0.308232, 0.345387, 0.384878, 0.426651, 0.470634, 0.516739, 0.564859,
 0.61487, 0.666632, 0.719988, 0.774767, 0.830782, 0.887834, 0.94571, 1.00419,
 1.06303, 1.12199, 1.18083, 1.23928, 1.29709, 1.35398, 1.4097, 1.46398,
 1.51656, 1.56717, 1.61557, 1.66151, 1.70475, 1.74507, 1.78225, 1.81609,
 1.84641, 1.87305, 1.89585, 1.91468, 1.92945, 1.94006, 1.94645, 1.94858}
```

The linear interpolation function is defined and the x[ϕ] function is redefined from it

```
func = Interpolation[interpolationData, InterpolationOrder → 1];
x[ϕ_] := s0 * H * func[ϕ]
τ[t_] := Mod[t + T / 2, T] - T / 2
s[t_] := x[Abs[ω τ[t]]]
```

```
Plot[s[t T] * 100, {t, - 1 / 2, 1 / 2}, PlotRange → All, Frame → True, FrameLabel →
  {"time (* T)", "Position of sheath edge\n from the plasma, s[t] (cm)"}]
```



The sheath size when fully extended is the value of x at $\phi=\pi$, or, the same s at time t=T/2,

```
sm = x[π]
sm = s[T / 2]
```

```
0.0121404

0.0121404
```

Note that in our model we redefine the one dimension variable x (where x = 0 is the plasma edge) to variable z (where z = 0 is the electrode). The sheath positiion z as a function of time (z = sm - x); where sm is the size of the fully extended sheath.

z[t] is defined below and plotted using meter and time in multiples of the radio-frequency period (T) units. Note that in the case of the z function, the sheath is fully extended at time t=0

```
z[t_] := s[T / 2] - s[t ]
Plot[z[t T ], {t, - 1, 1}, PlotRange → All,
 Frame → True, FrameLabel → {"time (* T)", "z(t) (m)"}]
```



Inverse of function x(ϕ): ϕ(x)

The inverse of the x(ϕ) function is required to calculate the electric field and to estimate the space potential across the sheath. For this we can use the numerical values obtained before but change the data columns so that ϕ would be a function of x. Use the data generated using the integration function (i.e., trapezoidal)

```
intfunc =
  Interpolation[Transpose[{integralfuncdata, zetas}], InterpolationOrder → 1];
ϕx[x_] := intfunc[x/(H * s0)]

Plot[ϕx[x], {x, 0, sm}, PlotRange → All, Frame → True, FrameLabel → {"x", "ϕ"}]
```



Electric Field in the sheath (z,t)

From Lieberman, 1989. Equation (??), where in the model of this document the relationship between the x variable and z variable is: x = sm - z. Therefore the electric field is not null for the condtion that s(t) < x = sm-z.

```
Ef[xx_, t_] := If[s[t] < xx, J/(ϵ0 ω) (Cos[ω t] - Cos[ϕx[xx]]), 0]
Ez[zz_, t_] := -Ef[sm - zz, t]
```

The graph below shows a plot of the electric field in the sheath from time -T/2 to time T/2,

The plot shows that the electric field is approximately zero where the position is behind the sheath edge as defined in the function.

```
Plot3D[Ez[zz / 100, tt * T], {zz, 0, 100 * sm}, {tt, -1 / 2, 1 / 2}, PlotRange → All,
 AxesLabel → {"z (m)", "t (* T)", "Ef (V/m)"}, Mesh → False, PlotPoints → {25, 25}]
```



Potential across the sheath (z,t)

The voltage or space potential across the sheath as a function of time can be obtained by integrating the electric field function as follows.

```
Potential[zz_, t_] := -NIntegrate[Ez[zzz, t],
   {zzz, 0, zz}, Method → "TrapezoidalRule", PrecisionGoal → 2]
```

The voltage integration is computationally expensive. Therefore, to illustrate the potential, we first generate a data set.

```
potential3D =
  Table[Potential[zz, tt], {zz, 0, sm, sm / 10}, {tt, -T / 2, T / 2, T / 25}];
```

And then plot the data using ListPlot3D.

The plot shows that the voltage is zero at the electrode position (z=0) and that is modulates in the sheath as a function of time, where the maximum voltage difference across the sheath occurs at t=0 (remember that in the model of this document we have changed from x variable to z and that the sheath is fully extended at time t = 0).

```
ListPlot3D[potential3D, AxesLabel → {"t (au)", "z (au)", "P (V)"}, Mesh → False]
```



The space potential can be also plotted as single curves for given time values. The graph below shows said curves for t from 0 to T in increments of T/10.

```
ListPlot[
 Table[
  Table[{zz * 100, Potential[zz, tt]}, {zz, 0, sm, sm / 10}],
  {tt, 0, T / 2, T / 10}], Joined → True,
 Frame → True, FrameLabel → {"z (cm)", "Potential (V)"}]
```



Note that in this model, the plasma parameters including the maximum voltage across the sheath Vpkpk were used as input to define the last parameter, the plasma discharge current density J. Therefore, when the sheath is fully extended, at z=sm and t=0 the voltage difference should be equal to Vpkpk.

This is calculated below; however, due to simplified numerical calculations in the calculation of the function of ϕ made to reduce computational time, the expected potential differs from the input potential by a couple of percent. However, using a better integration algorithm for integrating function of ϕ gives much better agreement with the expected voltage.

```
Vpkpk
Potential[sm, 0]
Abs[(Vpkpk - Potential[sm, 0]) / Vpkpk] * 100 "%"
```

$$\text{Abs}\left[\frac{\text{Vpkpk} - \text{Potential}[\text{sm}, 0]}{\text{Vpkpk}}\right] * 100\ \text{"\%"}$$

2000

1999.8

0.0100037 %